| /* | |
| DHCP Chat Server | |
| A simple server that distributes any incoming messages to all | |
| connected clients. To use, telnet to your device's IP address and type. | |
| You can see the client's input in the serial monitor as well. | |
| Using an Arduino Wiznet Ethernet shield. | |
| THis version attempts to get an IP address using DHCP | |
| Circuit: | |
| * Ethernet shield attached to pins 10, 11, 12, 13 | |
| created 21 May 2011 | |
| modified 9 Apr 2012 | |
| by Tom Igoe | |
| modified 02 Sept 2015 | |
| by Arturo Guadalupi | |
| Based on ChatServer example by David A. Mellis | |
| */ | |
| // Enter a MAC address and IP address for your controller below. | |
| // The IP address will be dependent on your local network. | |
| // gateway and subnet are optional: | |
| byte mac[] = { | |
| 0x00, 0xAA, 0xBB, 0xCC, 0xDE, 0x02 | |
| }; | |
| IPAddress ip(192, 168, 1, 177); | |
| IPAddress myDns(192, 168, 1, 1); | |
| IPAddress gateway(192, 168, 1, 1); | |
| IPAddress subnet(255, 255, 0, 0); | |
| // telnet defaults to port 23 | |
| EthernetServer server(23); | |
| boolean gotAMessage = false; // whether or not you got a message from the client yet | |
| void setup() { | |
| // You can use Ethernet.init(pin) to configure the CS pin | |
| //Ethernet.init(10); // Most Arduino shields | |
| //Ethernet.init(5); // MKR ETH shield | |
| //Ethernet.init(0); // Teensy 2.0 | |
| //Ethernet.init(20); // Teensy++ 2.0 | |
| //Ethernet.init(15); // ESP8266 with Adafruit Featherwing Ethernet | |
| //Ethernet.init(33); // ESP32 with Adafruit Featherwing Ethernet | |
| // Open serial communications and wait for port to open: | |
| Serial.begin(9600); | |
| while (!Serial) { | |
| ; // wait for serial port to connect. Needed for native USB port only | |
| } | |
| // start the Ethernet connection: | |
| Serial.println("Trying to get an IP address using DHCP"); | |
| if (Ethernet.begin(mac) == 0) { | |
| Serial.println("Failed to configure Ethernet using DHCP"); | |
| // Check for Ethernet hardware present | |
| if (Ethernet.hardwareStatus() == EthernetNoHardware) { | |
| Serial.println("Ethernet shield was not found. Sorry, can't run without hardware. :("); | |
| while (true) { | |
| delay(1); // do nothing, no point running without Ethernet hardware | |
| } | |
| } | |
| if (Ethernet.linkStatus() == LinkOFF) { | |
| Serial.println("Ethernet cable is not connected."); | |
| } | |
| // initialize the Ethernet device not using DHCP: | |
| Ethernet.begin(mac, ip, myDns, gateway, subnet); | |
| } | |
| // print your local IP address: | |
| Serial.print("My IP address: "); | |
| Serial.println(Ethernet.localIP()); | |
| // start listening for clients | |
| server.begin(); | |
| } | |
| void loop() { | |
| // wait for a new client: | |
| EthernetClient client = server.available(); | |
| // when the client sends the first byte, say hello: | |
| if (client) { | |
| if (!gotAMessage) { | |
| Serial.println("We have a new client"); | |
| client.println("Hello, client!"); | |
| gotAMessage = true; | |
| } | |
| // read the bytes incoming from the client: | |
| char thisChar = client.read(); | |
| // echo the bytes back to the client: | |
| server.write(thisChar); | |
| // echo the bytes to the server as well: | |
| Serial.print(thisChar); | |
| Ethernet.maintain(); | |
| } | |
| } | |